Modifications of the Two-treatment Design

We saw that with the two-treatment, two-period, two-sequence design, that we had to make assumptions concerning the carryover effect and treatment*period interaction. Using additional periods, sequences or both, we can improve the separability or even obtain complete separability of treatment and carryover effects.

Two Treatment, Two Sequences and Three Periods

There are 4 possible "dual balanced" designs for a two-treatment, two-sequence, three-period crossover study, where the dual sequences are interchanges of each other.

1.

	Period 1	Period 2	Period 3
Seq 1	A	B	B
Seq 2	B	A	A

2.

	Period 1	Period 2	Period 3
Seq 1	A	A	B
Seq 2	B	B	A

3.

	Period 1	Period 2	Period 3
Seq 1	A	B	A
Seq 2	B	A	B

4.

	Period 1	Period 2	Period 3
Seq 1	A	A	A
Seq 2	B	B	B

In Chapter 4 of "Cross-over Experiments" by Ratkowsky, Evans, and Allredge, 1993, show how the first design is superior to the other three designs. The design is the "fully-efficient" extra period design that we discussed in the previous section.

Two Treatment, Four Sequences and Two Periods

Balaam's design is a special design having t^2 sequences. For the two treatment case, this would mean four sequences. The only four possible sequences for the two treatment case would be the following:

	Period 1	Period 2
Sequence 1	A	B
Sequence 2	B	A
Sequence 3	A	A
Sequence 4	B	B

This design allows the simultaneous estimation of the carry-over effects with the treatment effects.

We will go through the example presented in the text. This placebo-controlled study described by Hunter et al. (1970) was designed to determent the effect of Amantadine on subjects suffering from Parkinsonism. There was a baseline period and two four-weekly treatment periods. Weekly scores of 11 physical signs were recorded and compiled. The following is the SAS code for the data:

```
data a;
input patient sequence period treat $ carry $ base y;
cards;
11 1 1 a n 14 12.50
11 1 2 a a 14 14.00
12 1 1 a n 27 24.25
12 1 2 a a 27 22.50
13 1 1 a n 19 17.25
13 1 2 a a 19 16.25
14 1 1 a n 30 28.25
14 1 2 a a 30 29.75
21 2 1 b n 21 20.00
21 2 2 b b 21 19.51
22 2 1 b n 11 10.50
22 2 2 b b 11 10.00
23 2 1 b n 20 19.50
23 2 2 b b 20 20.75
24 2 1 b n 25 22.50
24 2 2 b b 25 23.50
31 3 1 a n 9 8.75
31 3 2 b a 9 8.75
32 3 1 a n 12 10.50
32 3 2 b a 12 9.75
33 3 1 a n 17 15.00
33 3 2 b a 17 18.50
34 3 1 a n 21 21.00
34 3 2 b a 21 21.50
41 4 1 b n 23 22.0
41 4 2 a b 23 18.00
42 4 1 b n 15 15.00
42 4 2 a b 15 13.00
43 4 1 b n 13 14.00
43 4 2 a b 13 13.75
44 4 1 b n 24 22.75
44 4 2 a b 24 21.50
45 4 1 b n 18 17.75
45 4 2 a b 18 16.75
;
```

The carry variable is coded as “n” for the first period results and coded as the first period treatment for the second period results. We’ll now run four models which are presented in the following table:

Model	Description
1	Ignoring carry-over and treating patients as fixed
2	Ignoring carry-over and treating patients as random
3	Including carry-over and treating patients as fixed
4	Including carry-over and treating patients as random

The following SAS code is for running the analyses with the four models. Note that with Models 3 and 4, we drop the period term from the model and include the carry term instead, because period and carry-over are confounded (period 1 is always "n").

```
*patient fixed ignoring carryover;
proc mixed data=a;
class period patient treat;
model y= period treat patient;
lsmeans treat/pdiff;
run;

*patient random ignoring carryover;
proc mixed data=a asycov;
class period patient treat;
model y= period treat ;
random patient;
lsmeans treat/pdiff;
run;

*patient fixed including carryover;
proc mixed data=a;
class period patient treat base carry;
model y=treat patient carry;
lsmeans treat carry/pdiff;
run;

*patient random including carryover;
proc mixed data=a asycov;
class period patient treat base carry;
model y= treat carry ;
random patient;
lsmeans treat/pdiff;
run;
```

The following is selected output from Model 4.

```
Covariance Parameter
      Estimates

Cov Parm     Estimate

patient       30.4940
Residual       1.1193
```

```
          Asymptotic Covariance Matrix of Estimates

           Row    Cov Parm        CovP1       CovP2

             1    patient        121.11    -0.08013
             2    Residual     -0.08013      0.1774

              Differences of Least Squares Means

                                                          Standard
Effect  treat  carry  _treat  _carry  Estimate     Error    DF

treat   a             b               -1.2821    0.6977    14
carry          a              b        -0.1026    1.0128    14
carry          a              n        -0.1989    0.6245    14
carry          b              n       -0.09631    0.6229    14

              Differences of Least Squares Means

   Effect  treat  carry  _treat  _carry  t Value   Pr > |t|

   treat   a             b                -1.84     0.0875
   carry          a              b         -0.10     0.9208
   carry          a              n         -0.32     0.7548
   carry          b              n         -0.15     0.8793
```

The standard errors for the patient and residual variance components are the square roots of the diagonal elements of the Asymptotic Covariance Matrix of Estimates (121.11 for patient and 0.1774 for residual). The carry-over difference is the difference between "carry A" and "carry B", which is –0.1026. Note that carry-over is not significant in this study. Some statisticians may go ahead and drop this term from the model. However, the Balaam design was used in this study in order to be able to estimate both the treatment and carry-over effects which gives strong support for keeping carry-over in the model.

The following table summarizes the results from the four analyses:

		Fixed Patients	**Random Patients**
Ignoring carry-over	**Variance Components**		
	Patients	----	30.3 (10.91)
	Residual	1.05	1.1 (0.38)
	Treatment Difference (A-B)	-1.29 (0.49)	-1.24 (0.48)
Including carry-over	**Variance Components**		
	Patients	----	30.5 (11.0)
	Residual	1.12	1.1 (0.42)
	Treatment Difference	-1.42 (0.73)	-1.28(0.70)
	Carry-over Difference	-0.25 (1.06)	-0.10 (1.01)